United States
Environmental Protection
Agency

Environmental Sciences Research
Laboratory
Research Triangle Park NC 27711

EPA 600 2 79 140
August 1979

Research and Development

EPA

Implementation of a Microcomputer-Modified Electrical Aerosol Analyzer

RESEARCH REPORTING SERIES

Research reports of the Office of Research and Development, U.S. Environmental Protection Agency, have been grouped into nine series. These nine broad categories were established to facilitate further development and application of environmental technology. Elimination of traditional grouping was consciously planned to foster technology transfer and a maximum interface in related fields. The nine series are:

1. Environmental Health Effects Research
2. Environmental Protection Technology
3. Ecological Research
4. Environmental Monitoring
5. Socioeconomic Environmental Studies
6. Scientific and Technical Assessment Reports (STAR)
7. Interagency Energy-Environment Research and Development
8. "Special" Reports
9. Miscellaneous Reports

This report has been assigned to the ENVIRONMENTAL PROTECTION TECHNOLOGY series. This series describes research performed to develop and demonstrate instrumentation, equipment, and methodology to repair or prevent environmental degradation from point and non-point sources of pollution. This work provides the new or improved technology required for the control and treatment of pollution sources to meet environmental quality standards.

This document is available to the public through the National Technical Information Service, Springfield, Virginia 22161.

EPA-600/2-79-140
August 1979

IMPLEMENTATION OF A MICROCOMPUTER-MODIFIED

ELECTRICAL AEROSOL ANALYZER

by

Charles W. Lewis
Atmospheric Chemistry and Physics Division
Environmental Sciences Research Laboratory
Research Triangle Park, North Carolina 27711

ENVIRONMENTAL SCIENCES RESEARCH LABORATORY
OFFICE OF RESEARCH AND DEVELOPMENT
U.S. ENVIRONMENTAL PROTECTION AGENCY
RESEARCH TRIANGLE PARK, NORTH CAROLINA 27711

DISCLAIMER

This report has been reviewed by the Environmental Sciences Research Laboratory, U.S. Environmental Protection Agency, and approved for publication. Mention of trade names or commercial products does not constitute endorsement or recommendation for use.

ABSTRACT

A commercial electrical aerosol analyzer has been modified through addition of a digital voltmeter, a microcomputer and a printer. In unattended operation the system can acquire a differential particle size distribution, print and plot the resulting spectrum in any of three possible representations, and repeat the cycle at preselected intervals.

Details of the system assembly and instructions for its operation are given. The appendices contain a flow chart and documented listing of Program EAA, the microcomputer program which operates the system.

This reports covers a period from March 1978 to March 1979, and work was completed as of March 31, 1979.

CONTENTS

FIGURES

TABLES

ACKNOWLEDGEMENTS

The support given to this project by Robert K. Stevens, Chief, Inorganic Pollutant Analysis Branch, and the superior efforts of Ruth Barbour on a demanding manuscript are both happily acknowledged.

SECTION 1

INTRODUCTION

The electrical aerosol analyzer (EAA) has become a popular instrument for measuring submicron aerosol size distributions since its commercial introduction in 1967. By exploiting the known dependence of electrical mobility on particle size, such instruments permit the real time measurement of aerosol particle size over at least 2 decades of the range below 1 μm. While the importance of the EAA to aerosol studies is well recognized, certain features of the EAA remain to be improved. Of particular concern in the present commercial version (Model 3030, TSI, Inc.) is the need for an operator to manually record the sequence of electrical currents produced during the instrument's measurement cycle, and then to perform a series of arithmetical steps which convert the currents to the corresponding aerosol size spectrum.

To solve this limitation the EAA could be supplemented with a minicomputer and related peripherals to perform data input, reduction and output functions. Such a system, however, is relatively expensive and less portable than may be desired. In the following we show how the same objective can be accomplished at much less expense by means of a microcomputer-based addition to the EAA. The addition consists of unmodified commercial components, including a $180 microcomputer, a digital voltmeter, a teletypewriter or equivalent, and an optional audio cassette recorder. (A well-equipped laboratory may already possess some of these instruments.) In unattended operation the resulting system is able to (1) acquire a differential size distribution over 10 intervals of equal logarithmic size between 0.0032 and 1 μm diameter; (2) print and plot the distribution directly in a number, surface or volume representation; and (3) repeat the cycle indefinitely at a selectable frequency.

A condensed description of the system described in this report is available elsewhere (Lewis 1979). The material in Sections 3 and 4 and in the Appendices of the present report has not been published previously, and contains those details required to actually implement and operate the microcomputer-modified system.

SECTION 2

SYSTEM DESCRIPTION

HARDWARE

A block diagram of the assembled system is shown in Figure 1. The unmodified EAA consists of the two components shown within the dashed box -- the aerosol analyzer itself and its control module. In operation the control module causes the analyzer to generate a fixed time sequence of internal electric fields during a 2-min cycle. Aerosol entering the analyzer first becomes electrically charged and then is subjected to the particular electric field present at the time of passage. Depending on the magnitude of the electric field, only those aerosol particles larger than a minimum size are able to pass through the analyzer and contribute to a measurable aerosol current ("analyzer current"). Normally the control module is the component that measures the analyzer current sequence (whose values constitute the basic data output of the EAA) and presents the measured values visually to the operator. In the modified system, however, the current measuring and recording functions are performed by the additional components shown in Figure 1 outside the dashed box. Each of these components is described below.

Digital Voltmeter

The analyzer currents, available at an external coaxial connector on the EAA, are fed directly to the input of a laboratory digital voltmeter (DVM). (Although the signal of interest is referred to as a current, the EAA actually converts it internally to a voltage.) The DVM is conventional in all respects, but it must have the following capabilities (1) a resolution of at least 3½ digits, with a full scale reading of at least ±1 volt, and (2) a binary coded decimal (BCD) output, generating normal TTL positive-true voltage levels. The BCD output is a common interface type provided by some DVM manufacturers as an option for their instruments. Unfortunately, the DVM used in the EAA control module does not have this capability, resulting in the need for the additional DVM discussed here. Because of availability, the particular instrument used in this study was the 5½ digit multimeter Cimron Model DMM51 (California Instruments, Inc.).

Microcomputer

A KIM-1 microcomputer (Commodore/MOS, Inc.) was used to control the following: acquisition of analyzer currents from the DVM, arithmetical manipulation of the current values, and output of the computed aerosol



Figure 1. Aerosol system block diagram.

size distribution spectrum. The KIM-1 comprises all the elements of a full-fledged computer (a central processing unit, memory, and input/output interface circuitry), assembled on a single 21- by 26-cm printed circuit board. Details of the microcomputer have been described in earlier work (Lewis and Lamothe 1978).

Data Terminal

A serial 20-mA current loop interface is an integral part of the KIM-1 microcomputer. Thus, the most obvious choice for an output device is a conventional teletypewriter because of its frequent availability in laboratories. However, newer teletypewriter "equivalent" devices are attractive on the basis of competitive cost, compactness, quietness, and higher speed operation. (The microcomputer will operate up to at least 1200 baud.) One device that has proved satisfactory in operation with the present system is the Model 743 data terminal (Texas Instruments, Inc.). Any such teletypewriter "equivalent" device should be chosen with a "mark" parity (rather than either "even" or "odd" parity) option for compatibility with the KIM-1 microcomputer.

Newer data terminals are more commonly equipped with an RS-232-C interface rather than the older 20-mA current loop interface. A data terminal with an RS-232-C interface can also be used in the system, if a current loop/RS-232-C converter is inserted between the KIM-1 and the data terminal. One such converter which has been used without problems is the Model 312A 0566 TTY port interface adapter (United Data Services Company, Inc.)

Audio Cassette Recorder

The KIM-1 microcomputer program which operates the system of Figure 1 is stored in volatile memory. This means that whenever the microcomputer power is switched off the program is lost and must be reloaded before the next use. The only function of the audio cassette recorder is to provide a convenient means of performing the load operation. Many inexpensive cassette recorders exist that are suitable for this use. A cassette recorder is unnecessary, of course, if a teletypewriter with paper tape punch and reader is chosen as the data terminal, since the program may then be loaded from paper tape.

SOFTWARE

A flow diagram and listing of the microcomputer program (Program EAA) which controls the operation of the system are included as Appendices. The program is organized into three segments: acquisition, transformation, and output -- each of which is described below.

Acquisition

The acquisition segment transfers a DVM reading to the microcomputer memory upon command. The EAA control module sends the command via its "Read" signal, which is received on the Interrupt Request line of the microcomputer. This causes the "Data Ready" line from the BCD output of the DVM to be repeatedly scanned by the microcomputer until the appearance of a signal on this line indicates that a new DVM measurement value is available. (Typical DVM measurement frequencies are several per second.) Fourteen data bits, representing 3½ binary coded digits and algebraic sign, are then immediately transferred to the microcomputer memory for storage, and the microcomputer awaits the next "Read" signal. The end of the EAA measurement cycle is recognized by the number of "Read" signals which have occurred.

Transformation

Upon completion of an EAA cycle, the transformation segment of the program performs the same set of computations that an operator would normally perform to calculate the aerosol size distribution. That is, successive stored readings are subtracted, and the differences are multiplied by a set of pre-stored transformation constants. Any one of three sets of constants may be selected, corresponding to the aerosol size distribution expressed in a number, surface, or volume representation. The transformation constants given in Table 1 were taken from the (inverse) sensitivities for monodisperse aerosols tabulated by Liu et al. (1976). This tabulation is a "smoothed" version of earlier sensitivity data given by Liu and Pui (1975) and judged to be a slightly more realistic calibration than that provided by the earlier unadjusted data. The values in Table 1 and those of Liu et al. (1976) differ by a factor 4, because the quantity $\Delta \log D$ is incorporated into the values of Table 1.

During the subtraction step, two cases of invalid data are recognized by the program: (1) an overrange reading from the DVM, and (2) a DVM reading larger than the preceding one. In either case, the content of the affected size intervals will be arbitrarily set to zero by the program. Since the effect of an overrange condition on the BCD output will vary with the manufacturer, a programming change may be necessary, depending on the DVM used. Similarly, some manufacturers may reverse the normal convention that a positive DVM reading sets the sign bit of the positive-true BCD output. The simple program modifications which will accommodate either of these circumstances are discussed in Section 3.

Output

The transformed data are printed on the teletypewriter (or equivalent) in a table that lists the 10 values of average particle diameter and the value of the aerosol number, surface, or volume (in 4-place, floating-point, decimal form) corresponding to each diameter. The table is preceded by a character (N, S, or V) which identifies the choice of representation for the aerosol size distribution. A histogram plot can also be printed simultaneously with the

TABLE 1. TRANSFORMATION FACTORS FOR NUMBER, SURFACE, AND VOLUME REPRESENTATION OF EAA MEASUREMENTS*

D (μm)	$\Delta N/\Delta \log D \Delta v$ (cm^{-3} $volt^{-1}$)	$\Delta S/\Delta \log D \Delta v$ (μm^2 $cm^{-3} volt^{-1}$)	$\Delta V/\Delta \log D \Delta v$ (μm^3 $cm^{-3} volt^{-1}$)
0.00422	0.130×10^{10}	0.720×10^5	0.504×10^2
0.00750	0.344×10^8	0.608×10^4	0.760×10^1
0.0133	0.140×10^7	0.780×10^3	0.173×10^1
0.0237	0.716×10^6	0.127×10^4	0.500×10^1
0.0422	0.368×10^6	0.206×10^4	0.145×10^2
0.0750	0.189×10^6	0.334×10^4	0.417×10^2
0.133	0.972×10^5	0.540×10^4	0.120×10^3
0.237	0.496×10^5	0.880×10^4	0.346×10^3
0.422	0.255×10^5	0.143×10^5	0.100×10^4
0.750	0.131×10^5	0.231×10^5	0.289×10^4

*N = number, S = surface, V = volume, D = diameter

numerical results as a visual aid. As an example, Figure 2 shows the result of a volume distribution measurement performed on indoor aerosol in our laboratory.

With a teletypewriter operating at only 10 characters/sec, a printout generally cannot be completed before the first DVM reading must be taken in the next EAA cycle. The interrupt structure of the program solves this problem automatically by temporarily suspending output until the DVM reading has been taken.

Data are accumulated automatically and continuously from one EAA cycle after another by the microcomputer program. The operator, however, can choose beforehand to print only one of every N cycles, where N is variable between 1 and 255. Thus, a frequency of measurement is possible from once per 2 min to once per 8½ h. The means by which any of the output options may be selected are discussed in Section 4.

V		
.004	.0000	00♦
.008	.0084	00♦
.013	.0512	00♦
.024	.5871	00♦
.042	.4658	00♦
.075	.3410	01♦♦♦♦
.133	.2332	02♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦
.237	.3224	02♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦
.422	.2307	02♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦
.750	.1861	02♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦♦

Figure 2. Result of an ambient indoor aerosol measurement, illustrating tabular and histogram outputs. The first column lists the average particle diameter (in μm), while the second and third columns give the corresponding volume distribution values ΔV/ΔlogD (in $\mu m^3/cm^3$) as a fraction and power of 10, respectively.

SECTION 3

SYSTEM ASSEMBLY

ELECTRICAL CONNECTIONS

Details of the connections between the KIM-1 microcomputer and the other system components are listed in Table 2. The microcomputer and DVM are joined by a 16-conductor connection (14 data inputs for 3½ binary coded digits and polarity, a "data ready" input, and a ground line). The microcomputer and EAA control module are joined by a connection between the Interrupt Request input of the former and the "Read" terminal of the latter, in addition to a ground connection. (It is essential that the internal jumper of the EAA control module be set to the FAL position to provide a negative-true logic signal at the "Read" terminal.) Two 4-line connections link the data terminal and audio cassette recorder (optional) to the appropriate microcomputer ports. All connections are simple passive ones, since all active circuitry necessary for the microcomputer to transfer information to and from the indicated devices is an integral part of the microcomputer itself. The only connection which does not involve the microcomputer is a coaxial cable from the "Analyzer Current" connector of the EAA to the input of the DVM.

PROGRAM MODIFICATIONS

Up to three modifications of Program EAA may be required, depending on the baud requirement of the data terminal and the overrange and polarity characteristics of the DVM. Although such changes can be made each time after the program is loaded but before its execution has been started, it is more convenient for the user to produce a new program tape with the changes incorporated permanently. The user should consult the documentation supplied with the KIM-1 for the procedures to generate new paper or magnetic tapes.

Data Terminal Baud

Program EAA causes the bit rate (baud) for the system's output device to be set at the beginning of the program's execution. In the current version of the program, the baud is set to 300 (30 characters/sec) in this step. It should be appreciated that any baud setting entered in addresses 17F2 and 17F3 before program execution has begun will be overwritten at this time. If a baud setting different from 300 is required for a specific data terminal, the contents of two KIM-1 addresses must be altered according to Table 3.

TABLE 2. INPUT/OUTPUT CONNECTIONS

Microcomputer connector pin	I/O port designation	I/O signal
A-A		+5 VDC
A-14	PA0	1 BIT†
A-4	PA1	2 BIT†
A-3	PA2	4 BIT†
A-2	PA3	8 BIT†
A-5	PA4	1 BIT
A-6	PA5	2 BIT
A-7	PA6	4 BIT
A-8	PA7	8 BIT
A-9	PB0	1 BIT
A-10	PB1	2 BIT
A-11	PB2	4 BIT
A-12	PB3	8 BIT§
A-13	PB4	1 BIT§
A-16	PB5	POLARITY (DVM)
A-15	PB7	DATA READY (DVM)
E-4	IRQ	READ (EAA CONTROL MODULE)
A-R		TTY KEYBOARD (+)
A-S		TTY PRINTER (+)
A-T		TTY KEYBOARD (-)
A-U		TTY PRINTER (-)
A-1*		GROUND (FOR DVM, EAA READ, AUDIO RECORDER, +5 AND +12VDC)
	OPTIONAL:	
A-L		AUDIO IN (EARPHONE JACK)
A-M		AUDIO OUT (MICROPHONE JACK)
A-N		+12 VDC

*A jumper wire must connect pins A-1 and A-K. Also if the teletypewriter keyboard, instead of the microcomputer keyboard, is used as the input device a jumper wire must connect pins A-21 and A-V.
†Least significant BCD digit from DVM.
§Most significant BCD digit from DVM

TABLE 3. PROGRAM ALTERATIONS TO CHANGE BAUD

	Baud						
	110	300*	600	1200	2400	4800	9600
Address							
0105	80	E9	74	38	1A	06	03
010A	02	00	00	00	00	00	00

*default

In principle the KIM-1 should be able to adjust itself to the requirements of an arbitrary data terminal through a "RESET/RUBOUT" sequence, as described in the KIM-1 documentation. This feature has not proven reliable, however, and hence the alternative procedure given above.

DVM Overrange

Program EAA recognizes an overrange condition of the DVM by comparing the two leading digits of every DVM reading to a prestored characteristic overrange pattern. In the current version of the program, the pattern is "12", corresponding to the DVM overrange reading, "1.200 volts". Other common overrange conditions are 1.999 and 0.000, depending on the DVM manufacturer. For Program EAA to function correctly, the proper two digit pattern for the particular DVM must be entered at address 00E5.

DVM Polarity

Program EAA assumes that a positive signal input to the DVM will set the sign bit of its positive-true output. In the unlikely event that the DVM operates on an opposite convention for the sign bit, the contents of three addresses must be altered, as given in Table 4.

TABLE 4. PROGRAM ALTERATIONS FOR CASE OF UNCONVENTIONAL SIGN BIT

Address	Normal Content	Altered Content
0054	F0	D0
0067	D0	F0
0089	D0	F0

SECTION 4

OPERATING INSTRUCTIONS

PROCEDURE

1. Power on all system components (EAA, EAA Control, DVM, KIM-1 and Data Terminal).

2. Load Program EAA into KIM-1 from paper tape or audio cassette tape:

   A. Paper Tape

      1. Set teletypewriter to LINE position.

      2. Set teletypewriter baud to 110 by setting KIM-1 address 17F2 = 80 and address 17F3 = 02.

      3. Switch from KIM-1 keyboard to teletypewriter by connecting pins A-21 and A-V, and observe "KIM XXXX XX" to be printed by teletypewriter.

      4. Place leader portion of paper tape in reader. Type L. Move reader switch to START position to begin reading tape.

      5. When the end of the tape is reached, "KIM XX XXXX" will be printed if the read operation has been successful.

      6. Disconnect pins A-21 and A-V to reactivate KIM-1 keyboard.

   B. Audio Cassette Tape

      1. Plug Audio In connector from KIM-1 into Earphone socket of audio cassette recorder. Make certain that Microphone socket is unoccupied. (Conversely, when recording, check that Earphone socket is unoccupied.) Rewind and adjust volume control to about half scale.

      2. Set address 00F1 = 00 and address 17F9 = previously chosen identification number for the program.

    3. Select address 1873, and depress KIM-1 "GO" key.

    4. Depress Play button on recorder. When the KIM-1 display relights showing "0000 xx," the program has been successfully loaded. Disconnect recorder.

3. For the EAA, adjust all flow rates and check Ionizer Current, Ionizer Voltage and switch in Externally Programmed position.

4. If the DVM has both auto ranging and manual operation modes, select manual mode. Select the "1 Volt DC" scale of the DVM and position the decimal point to the right of the left-most digit (i.e., X.XXX)

5. Depress "CONT. RUN", "ALL", "CURRENT", and "RESET" buttons on EAA Control module.

6. Modify the KIM-1 program according to which output features are desired. (See USER OPTIONS below.) Make any further program changes required. (See SYSTEM ASSEMBLY.)

7. Select address 0100, and depress the KIM-1 "GO" key to begin program execution.

8. Depress "START/STEP" button on EAA Control module.

9. To stop program execution depress the KIM-1 "ST" key.

10. To restart program execution depress "RESET" button on EAA Control module and repeat steps 7 and 8.

USER OPTIONS

Number, Surface and Volume Representations

The contents of addresses 0022 and 0023 determine the output form of the size distribution, as given in Table 5. The output will be preceded by the letter "V", "N", or "S", identifying volume, number, and surface representations, respectively.

TABLE 5. PROGRAM ALTERATIONS FOR CHOICE OF OUTPUT REPRESENTATION

Address	Content	Output Choice
0022	FE	volume*
0023	02	dV/DlogD, $\mu m^3/cm^3$
0022	F6	number
0023	02	dN/dlogD, $1/cm^3$
0022	06	surface
0023	03	dS/dlogD, $\mu m^2/cm^3$

*default

Output Frequency

The frequency of printed output may be controlled by the content of address 00E4, as given in Table 6. Note that the output frequency is specified as a hexadecimal number. Thus for example if one wishes to output every twentieth cycle, address 00E4 should contain 14 ($14_{16} = 20_{10}$).

TABLE 6. PROGRAM ALTERATION FOR CHOICE OF OUTPUT FREQUENCY

Address	Content	Effect
00E4	N_{16}^{*}	output occurs once per N_{16} EAA cycles

*default = 01 (output every cycle)

uppress Plotting

Histogram plotting to accompany the tabular output of the aerosol size istribution may be selected or omitted, according to the content of addresses 3B4, 03B5 and 03B6, as given in Table 7.

TABLE 7. PROGRAM ALTERATION TO SUPRESS HISTOGRAM PLOTTING

Address	Content	Effect
03B4	20	Plotting selected*
03B5	D0	
03B6	03	
03B4	EA	Plotting suppressed
03B5	EA	
03B6	EA	

*default

REFERENCES

Lewis, C.W. 1979. A Microcomputer-Modified Electrical Aerosol Analyzer. J. Aerosol Sci. (to be published).

Lewis, C.W. and P.J. Lamothe. 1978. A Microcomputer-Modified Particle Size Spectrometer. J. Aerosol Sci. 9:391-397.

Liu, B.Y.H. and D.Y.H. Pui. 1975. On the Performance of the Electrical Aerosol Analyzer. J. Aerosol Sci. 6:249-264.

Liu, B.Y.H., D.Y.H. Pui, and A. Kapadia. 1976. Electrical Aerosol Analyzer: History, Principle and Data Reduction. Particle Technology Laboratory Publication No. 303, University of Minnesota, Minneapolis.

APPENDIX A

PROGRAM EAA FLOW CHART



PROGRAM EAA (2)



PROGRAM EAA (3)




OUTPUT
INITIALIZE
LINE
PRINT NEXT SIZE BIN AND CONTENT
SUBROUTINE PLOT ADJUST BIN CONTENT TO HAVE EXPONENT = EXP AND PLOT RESULTING HI ORDER BYTE
DONE
NO
LINE
YES
MAIN

PROGRAM EAA (5)



APPENDIX B

PROGRAM EAA LISTING

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0000	A5 2B	MAIN	LDA STEP	MAIN WAITING LOOP
0002	C9 16		CMP #16	
0004	D0 FA		BNE MAIN	END OF ACQUISITION CYCLE?
0006	A9 00		LDA #00	YES
0008	85 2B		STA STEP	RESET STEP = 0
000A	C6 29		DEC SKIP	
000C	D0 F2		BNE MAIN	CYCLE TO BE NEGLECTED?
000E	A5 E4		LDA FREQ	NO
0010	85 29		STA SKIP	RESET SKIP = FREQ
0012	A0 00		LDY #00	
0014	A9 1D		LDA #1D	
0016	85 2A		STA CHAN	CHAN = 29_{10}
0018	F8		SED	
0019	20 30 00	DEL	JSR DELTA	COMPUTE DIFFERENCE
001C	C0 14		CPY #14	
001E	D0 F9		BNE DEL	ALL DIFFERENCES COMPUTED?
0020	D8		CLD	YES, CLEAR DECIMAL MODE AND
0021	4C FE 02		JMP ENTR2	GO TO PRINT ASCII LETTER
0024	XX XX XX			UNUSED
0027	XX	TIMER		NINE TEMPORARY STORAGE
0028	XX	EXP		ADDRESSES
0029	XX	SKIP		
002A	XX	CHAN		
002B	XX	STEP		
002C	XX	MLO		
002D	XX	MHI		
002E	XX	NLO		
002F	XX	NHI		

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0030	C8	DELTA	INY	SUBTRACTION SUBROUTINE
0031	B9 1E 02		LDA DMM, Y	LOAD 1ST DMM HI ORDER BYTE
0034	C8		INY	
0035	C8		INY	
0036	25 E5		AND OFFSCL	
0038	C5 E5		CMP OFFSCL	
003A	D0 04		BNE NO	1ST DMM VALUE OVERRANGE?
003C	88		DEY	YES, STORE ZERO
003D	4C 90 00		JMP NUL	
0040	B9 1E 02	NO	LDA DMM, Y	NO, LOAD 2ND DMM HI ORDER
0043	88		DEY	BYTE
0044	25 E5		AND OFFSCL	
0046	C5 E5		CMP OFFSCL	
0048	D0 03		BNE NORMAL	2ND DMM VALUE OVERRANGE?
004A	4C 90 00		JMP NUL	YES, STORE ZERO
004D	88	NORMAL	DEY	NO, RESTORE Y
004E	B9 1E 02		LDA DMM,Y	
0051	88		DEY	
0052	24 E6		BIT SIGN	1ST DMM VALUE POSITIVE?
0054	F0 19		BEQ EXCHNG	NO, GO TO EXCHNG
0056	A2 00		LDX #00	YES, LOAD 1ST DMM VALUE
0058	B9 1E 02	RPT	LDA DMM, Y	INTO MLO AND MHI,
005B	95 2C		STA MLO, X	AND 2ND DMM VALUE
005D	C8		INY	INTO NLO AND NHI
005E	E8		INX	
005F	E0 04		CPX #04	
0061	D0 F5		BNE RPT	
0063	88		DEY	
0064	88		DEY	
0065	24 E6		BIT SIGN	2ND DMM VALUE POSITIVE?
0067	D0 22		BNE SUBTR	YES, GO TO SUBTR
0069	20 C0 00	ADD	JSR SUM	NO, CALC. ABS(M) + ABS (N)
006C	4C A3 00		JMP STORE	

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
006F	B9 1E 02	EXCHNG	LDA DMM, Y	LOAD 1ST DMM VALUE
0072	85 2E		STA NLO	INTO NLO AND NHI
0074	C8		INY	
0075	B9 1E 02		LDA DMM, Y	
0078	85 2F		STA NHI	
007A	C8		INY	
007B	B9 1E 02		LDA DMM, Y	LOAD 2ND DMM VALUE
007E	85 2C		STA MLO	INTO MLO AND MHI
0080	C8		INY	
0081	B9 1E 02		LDA DMM, Y	
0084	85 2D		STA MHI	
0086	88		DEY	
0087	24 E6		BIT SIGN	2ND DMM VALUE POSITIVE
0089	D0 05		BNE NUL	YES, GO TO NUL
008B	20 D1 00	SUBTR	JSR DIFF	NO, CALC. ABS(M) - ABS(N)
008E	B0 13		BCS STORE	POSITIVE RESULT?
0090	A6 2A	NUL	LDX CHAN	NO,
0092	A9 00		LDA #00	STORE ZERO
0094	9D 00 02		STA DMMDIF, X	
0097	CA		DEX	
0098	9D 00 02		STA DMMDIF, X	
009B	CA		DEX	
009C	9D 00 02		STA DMMDIF, X	
009F	CA		DEX	
00A0	86 2A		STX CHAN	
00A2	60		RTS	SUBROUTINE RETURN
00A3	A6 2A	STORE	LDX CHAN	YES,
00A5	A5 2D		LDA MHI	STORE RESULT:
00A7	9D 00 02		STA DMMDIF, X	HI ORDER BYTE
00AA	CA		DEX	
00AB	A5 2C		LDA MLO	
00AD	9D 00 02		STA DMMDIF, X	LO ORDER BYTE
00B0	CA		DEX	

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
00B1	A9 01		LDA #01	
00B3	9D 00 02		STA DMMDIF, X	EXPONENT
00B6	CA		DEX	
00B7	86 2A		STX CHAN	
00B9	60		RTS	SUBROUTINE RETURN
00BA	78	CR	SEI	SUBROUTINE FOR
00BB	20 2F 1E		JSR CRLF	CARRIAGE RETURN, LINE FEED
00BE	58		CLI	
00BF	60		RTS	SUBROUTINE RETURN
00C0	20 E1 02	SUM	JSR MASK	SUBROUTINE TO
00C3	18		CLC	CALC. ABS(M) + ABS(N)
00C4	A5 2C		LDA MLO	
00C6	65 2E		ADC NLO	
00C8	85 2C		STA MLO	
00CA	A5 2D		LDA MHI	
00CC	65 2F		ADC NHI	
00CE	85 2D		STA MHI	
00D0	60		RTS	SUBROUTINE RETURN
00D1	20 E1 02	DIFF	JSR MASK	SUBROUTINE TO
00D4	38		SEC	CALC. ABS(M) - ABS(N)
00D5	A5 2C		LDA MLO	
00D7	E5 2E		SBC NLO	
00D9	85 2C		STA MLO	
00DB	A5 2D		LDA MHI	
00DD	E5 2F		SBC NHI	
00DF	85 2D		STA MHI	
00E1	60		RTS	SUBROUTINE RETURN
00E2	XX			UNUSED
00E3	XX			UNUSED

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
00E4	01	FREQ		FREQUENCY OF EAA PRINTOUT
00E5	12	OFFSCL		DMM OVERRANGE MASK
00E6	20	SIGN		DMM SIGN BIT MASK
00E7	XX	TEMPA		TWO TEMPORARY STORAGE
00E8	XX	TEMPB		ADDRESSES
00E9	XX	PLIERPNTL/COUNTER		THREE POINTER ADDRESSES
00EA	02	PLIERPNTH		FOR MULTIPLY SUBROUTINE
00EB	XX	CANDPNTL		
00EC	02	CANDPNTH		
00ED	XX	PRODPNTL		
00EE	02	PRODPNTH		
00EF	XX XX XX			ADDRESSES 00EF-
00F2	XX XX XX			00FF NOT AVAILABLE FOR
00F5	XX XX XX			USER PROGRAMS
00F8	XX XX XX			
00FB	XX XX XX			
00FE	XX XX			

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0100	D8	START	CLD	STARTING ADDRESS, CLEAR
0101	A2 FF		LDX #FF	DECIMAL MODE
0103	9A		TXS	INITIALIZE STACK
0104	A9 E9		LDA #E9	SET BAUD RATE = 300
0106	8D F2 17		STA 17F2	
0109	A9 00		LDA #00	
010B	8D F3 17		STA 17F3	
010E	20 2F 1E		JSR CRLF	CARRIAGE RETURN, LINE FEED
0111	20 2F 1E		JSR CRLF	
0114	EA		NOP	
0115	A9 00		LDA #00	DEFINE PORTS A & B
0117	8D 01 17		STA 1701	AS ALL INPUTS
011A	8D 03 17		STA 1703	
011D	8D FA 17		STA 17FA	SET UP INTERRUPT
0120	A9 1C		LDA #1C	VECTOR FOR NMI
0122	8D FB 17		STA 17FB	
0125	A9 AC		LDA #AC	SET UP INTERRUPT
0127	8D FE 17		STA 17FE	VECTOR FOR IRQ
012A	A9 02		LDA #02	
012C	8D FF 17		STA 17FF	
012F	A5 E4		LDA FREQ	INITIALIZE SKIP
0131	85 29		STA SKIP	
0133	A9 00		LDA #00	
0135	85 2B		STA STEP	SET STEP = 0
0137	58		CLI	ENABLE IRQ
0138	4C 00 00		JMP MAIN	BEGIN DATA ACQUISITION

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
013B	48	FPM	PHA	DECIMAL FLOATING POINT
013C	08		PHP	MULTIPLY SUBROUTINE
013D	86 E7		STX TEMPA	(DECIMAL MODE
013F	84 E8		STY TEMPB	PREVIOUSLY SET)
0141	A9 03		LDA #03	
0143	8D E4 01		STA CNTSHIFT	INITIALIZE SHIFT COUNTER.
0146	A0 06		LDY #06	SET TO ZERO:
0148	A9 00		LDA #00	CANDSRH, CANDSRL,
014A	99 E7 01	AGN	STA CNTEXP,Y	PROD1, PROD2,
014D	88		DEY	PROD3, PROD4,
014E	10 FA		BPL AGN	CNTEXP.
0150	A0 02		LDY #02	LOAD PLIERH
0152	B1 E9	LDP	LDA (PLIERPNTL),Y	AND PLIERL
0154	88		DEY	INTO RESIDENT
0155	99 E5 01		STA PLIERL,Y	STORAGE
0158	D0 F8		BNE LDP	
015A	A0 01		LDY #01	LOAD CANDL
015C	B1 EB		LDA (CANDPNTL),Y	AND CANDH
015E	8D EF 01		STA CANDL	INTO RESIDENT
0161	C8		INY	STORAGE
0162	B1 EB		LDA (CANDPNTL),Y	
0164	8D EE 01		STA CANDH	
0167	AD E5 01	CKPLIER	LDA PLIERL	CHECK 4 LSB'S
016A	29 0F		AND #0F	OF PLIERL
016C	F0 13		BEQ DECCNTSHIFT	
016E	A8		TAY	Y = ADDITION COUNTER
016F	18	SETX	CLC	ADD ALL 4 BYTES
0170	A2 03		LDX #03	OF CAND TO PROD
0172	BD EC 01	SUM	LDA CANDSRH,X	
0175	7D E8 01		ADC PROD1,X	
0178	9D E8 01		STA PROD1,X	
017B	CA		DEX	

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
017C	10 F4		BPL SUM	
017E	88		DEY	REPEAT ADDITIONS
017F	D0 EE		BNE SETX	FOR Y TIMES.
0181	CE E4 01	DECCNTSHIFT	DEC CNTSHIFT	BRANCH IF ALL 4 DIGITS
0184	30 16		BMI LJD	OF PLIER HAVE BEEN USED
0186	A0 04		LDY #04	Y = 4 BIT COUNTER
0188	A2 03	NEXTDIG	LDX #03	X = 4 BYTE COUNTER
018A	18		CLC	SHIFT CAND
018B	3E EC 01	ROCAND	ROL CANDSRH,X	LEFT 1 BIT
018E	CA		DEX	REPEAT FOR ALL 4
018F	10 FA		BPL ROCAND	BYTES
0191	4E E6 01		LSR PLIERH	SHIFT PLIER
0194	6E E5 01		ROR PLIERL	RIGHT 1 BIT
0197	88		DEY	REPEAT FOR 4
0198	D0 EE		BNE NEXTDIG	BITS.
019A	F0 CB		BEQ CKPLIER	
019C	A0 00	LJD	LDY #00	
019E	B1 E9	ADDEXP	LDA (PLIERPNTL),Y	ADD EXPONENTS
01A0	18		CLC	OF PLIER AND
01A1	71 EB		ADC (CANDPNTL),Y	CAND
01A3	8D E7 01		STA CNTEXP	
01A6	A9 F0	CKMSD	LDA #F0	CHECK 4 MSB'S
01A8	2C E8 01		BIT PROD1	OF PROD
01AB	D0 1E		BNE STOREXP	BRANCH IF ≠ 0
01AD	A0 04		LDY #04	4 BITS FOR EACH BCD
01AF	18	DIGSHIFT	CLC	DIGIT
01B0	A2 03		LDX #03	PROD CONTAINS 4 BYTES
01B2	3E E8 01	BITSHIFT	ROL PROD1,X	SHIFT PROD LEFT
01B5	CA		DEX	1 BIT. REPEAT FOR
01B6	10 FA		BPL BITSHIFT	ALL 4 BYTES
01B8	88		DEY	REPEAT FOR 4 BITS
01B9	D0 F4		BNE DIGSHIFT	
01BB	38	DECEXP	SEC	DECREMENT EXPONENT

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENTS
01BC	AD E7 01		LDA CNTEXP	FOR EACH DIGIT
01BF	E9 01		SBC #01	SHIFTED
01C1	8D E7 01		STA CNTEXP	
01C4	D0 E0		BNE CKMSD	
01C6	F0 03		BEQ STOREXP	
01C8	XX XX XX			UNUSED
01CB	AD E7 01	STOREXP	LDA CNTEXP	(Y = 0)
01CE	91 ED		STA (PRODPNTL),Y	EXPONENT STORED AT LOWEST
01D0	A0 02		LDY #02	ADDRESS
01D2	AD E8 01		LDA PROD1	
01D5	91 ED		STA (PRODPNTL),Y	HI STORED AT HIGHEST ADDRESS
01D7	88		DEY	
01D8	AD E9 01		LDA PROD2	
01DB	91 ED		STA (PRODPNTL),Y	LO STORED AT MIDDLE ADDRESS
01DD	A6 E7		LDX TEMPA	
01DF	A4 E8		LDY TEMPB	
01E1	28		PLP	
01E2	68		PLA	
01E3	60		RTS	RETURN FROM SUBROUTINE
01E4	XX	CNTSHIFT		TWELVE TEMPORARY
01E5	XX	PLIERL		STORAGE ADDRESSES
01E6	XX	PLIERH		USED BY MULTIPLY
01E7	XX	CNTEXP		SUBROUTINE
01E8	XX	PROD1		
01E9	XX	PROD2		
01EA	XX	PROD3		
01EB	XX	PROD4		
01EC	XX	CANDSRH		
01ED	XX	CANDSRL		
01EE	XX	CANDH		
01EF	XX	CANDL		
01F0	XX XX XX			UNUSED
01F3	XX XX XX			UNUSED

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENTS
01F6	XX XX XX			UNUSED
01F9	XX XX XX			UNUSED
01FC	XX XX			UNUSED
01FE	XX			RESERVED FOR
01FF	XX	STACK		STACK OPERATION

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0200	XX XX XX	DMMDIF		ADDRESSES 0200-021D
.	. . .			CONTAIN COMPUTED
.	. . .			DIFFERENCES OF
.	. . .			SUCCESSIVE DMM READINGS
.	. . .			STORED AS 10_{10} THREE
.	. . .			BYTE (EXPONENT, LO
.	. . .			ORDER, HI ORDER)
021B	XX XX XX			BCD NUMBERS.
021E	XX XX	DMM		ADDRESSES 021E-0233
.	. .			CONTAIN SUCCESSIVE
.	. .			DMM READINGS FROM
.	. .			AN EAA CYCLE,
.	. .			STORED AS 11_{10} TWO
.	. .			BYTE (LO ORDER, HI
0232	XX XX			ORDER) BCD NUMBERS.

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0234	04 90 28	VOLUME		VOLUME TRANSFORMATION
0237	04 00 10			FACTORS FOR EACH SIZE
023A	03 60 34			BIN, STORED IN REVERSE
023D	03 00 12			DECIMAL FLOATING POINT
0240	02 70 41			FORMAT. FOR EXAMPLE,
0243	02 50 14			THE FIRST FACTOR IS
0246	01 00 50			0.2890×10^4
0249	01 30 17			
024C	01 00 76			
024F	02 40 50			
0252	05 10 23	SURFACE		SURFACE TRANSFORMATION
0255	05 30 14			FACTORS. THE
0258	04 00 88			FIRST FACTOR IS
025B	04 00 54			0.2310×10^5
025E	04 40 33			
0261	04 60 20			
0264	04 70 12			
0267	03 00 78			
026A	04 80 60			
026D	05 00 72			
0270	05 10 13	NUMBER		NUMBER TRANSFORMATION
0273	05 50 25			FACTORS. THE FIRST
0276	05 60 49			FACTOR IS
0279	05 20 97			0.1310×10^5
027C	06 90 18			
027F	06 80 36			
0282	06 60 71			
0285	07 00 14			
0288	08 40 34			
028B	10 00 13			

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
028E	30 35 37	TABLE		AVE. PARTICLE DIA.
0291	32 32 34			(IN NANOMETERS) FOR
0294	37 33 32			EACH SIZE BIN, STORED
0297	33 33 31			IN REVERSE ASCII FORMAT.
029A	35 37 30			FOR EXAMPLE, THE FIRST
029D	32 34 30			ENTRY CORRESPONDS TO 750
02A0	34 32 30			WITH 30 = 0, 35 = 5, 37 = 7
02A3	33 31 30			
02A6	38 30 30			
02A9	34 30 30			

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
02AC	48	IRQ	PHA	INTERRUPT BY EAA READ
02AD	8A		TXA	SAVE A
02AE	48		PHA	SAVE X
02AF	A6 2B		LDX STEP	X = STEP
02B1	AD 02 17	EXAM	LDA 1702	
02B4	10 FB		BPL EXAM	IS DMM READY?
02B6	AD 00 17		LDA 1700	YES, STORE 2 LSD'S
02B9	9D 1E 02		STA DMM, X	FROM DMM
02BC	E8		INX	
02BD	AD 02 17		LDA 1702	STORE 2 MSD'S
02C0	9D 1E 02		STA DMM, X	FROM DMM
02C3	E8		INX	
02C4	A9 FF		LDA #FF	DELAY FOR
02C6	20 D0 02		JSR WAIT	256 x 11.264 = 2884 MSEC.
02C9	86 2B		STX STEP	STEP = X
02CB	68		PLA	
02CC	AA		TAX	RESTORE X
02CD	68		PLA	RESTORE A
02CE	58		CLI	ENABLE IRQ
02CF	40		RTI	INTERRUPT RETURN

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
02D0	85 27	WAIT	STA TIMER	WAITING SUBROUTINE
02D2	A9 0A	AGAIN	LDA #0A	SET CLOCK FOR
02D4	8D 07 17		STA 1707	11 x 1.024 = 11.264 MSEC.
02D7	AD 07 17	FLAG	LDA 1707	REMAIN IN SUBROUTINE
02DA	F0 FB		BEQ FLAG	UNTIL (TIMER) X 11.264 MSEC
02DC	C6 27		DEC TIMER	HAS PASSED
02DE	D0 F2		BNE AGAIN	
02E0	60		RTS	SUBROUTINE RETURN
02E1	A5 2D	MASK	LDA MHI	MASKING SUBROUTINE
02E3	29 1F		AND #1F	SET 3 MSB'S OF
02E5	85 2D		STA MHI	MHI AND NHI TO
02E7	A5 2F		LDA NHI	ZERO
02E9	29 1F		AND #1F	
02EB	85 2F		STA NHI	
02ED	60		RTS	SUBROUTINE RETURN
02EE	78	CH	SEI	SUBROUTINE TO
02EF	20 A0 1E		JSR OUTCH	OUTPUT CHARACTER
02F2	58		CLI	
02F3	60		RTS	SUBROUTINE RETURN
02F4	XX XX			UNUSED

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
02F6	A9 70	ENTR1	LDA #70	MULTIPLY ROUTINE
02F8	85 E9		STA E9	LOAD PLIERPNTL WITH LO ORDER
02FA	A9 4E		LDA #4E	ADDRESS OF NUMBER
02FC	D0 0E		BNE CHAR	ASCII "N"
02FE	A9 34	ENTR2	LDA #34	LOAD PLIERPNTL WITH LO ORDER
0300	85 E9		STA E9	ADDRESS OF VOLUME
0302	A9 56		LDA #56	ASCII "V"
0304	D0 06		BNE CHAR	
0306	A9 52	ENTR3	LDA #52	LOAD PLIERPNTL WITH LO ORDER
0308	85 E9		STA E9	ADDRESS OF SURFACE
030A	A9 53		LDA #53	ASCII "S"
030C	20 A0 1E	CHAR	JSR OUTCH	PRINT "V" OR "S" OR "N"
030F	20 2F 1E		JSR CRLF	CARRIAGE RETURN, LINE FEED
0312	A9 00		LDA #00	INITIALIZE:
0314	85 EB		STA EB	CANDPNTL AND
0316	85 ED		STA ED	PRODPNTL
0318	A0 0A		LDY #0A	10_{10}SIZE BINS
031A	F8	CALL	SED	SET DECIMAL MODE
031B	20 3B 01		JSR FPM	MULTIPLY
031E	D8		CLD	CLEAR DECIMAL MODE
031F	88		DEY	
0320	D0 03		BNE #03	IF ALL BINS DONE
0322	4C 39 03		JMP SORT	GO TO SORT
0325	A2 04		LDX #04	IF NOT DONE,
0327	F6 E9	INCP	INC E9,X	INCREMENT
0329	F6 E9		INC E9,X	PLIERPNTL,
032B	F6 E9		INC E9,X	CANDPNTL, AND
032D	CA		DEX	PRODPNTL
032E	CA		DEX	THREE TIMES
032F	10 F6		BPL INCP	
0331	30 E7		BMI CALL	DO NEXT MULTIPLICATION

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0333	78	SP	SEI	SUBROUTINE TO
0334	20 9E 1E		JSR OUTSP	OUTPUT SPACE
0337	58		CLI	
0338	60		RTS	SUBROUTINE RETURN

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0339	A9 00	SORT	LDA #00	ROUTINE TO FIND
033B	85 28		STA EXP	LARGEST EXPONENT
033D	A2 1B		LDX #1B	$1B_{16}$ = 10 BINS X 3 - 3
033F	BD 00 02	TEST	LDA 0200,X	LOAD NEXT EXPONENT
0342	C5 28		CMP EXP	
0344	90 02		BCC DECR	STORE IF LARGER
0346	85 28		STA EXP	THAN PREVIOUS EXP'S
0348	CA	DECR	DEX	
0349	CA		DEX	
034A	CA		DEX	
034B	10 F2		BPL TEST	DONE?
034D	A2 1B		LDX #1B	YES, RESET X
034F	BD 00 02	TRY	LDA 0200,X	LOAD NEXT EXPONENT
0352	C5 28		CMP EXP	EXPONENT MAXIMUM?
0354	D0 10		BNE RDUCE	NO, GO TO RDUCE
0356	E8		INX	
0357	E8		INX	
0358	BD 00 02		LDA 0200, X	YES, LOAD HI ORDER
035B	CA		DEX	BYTE
035C	CA		DEX	
035D	C9 55		CMP #55	55_{10} = MAX. ORDINATE
035F	90 05		BCC RDUCE	MAX. ORDINATE EXCEEDED?
0361	E6 28		INC EXP	YES
0363	4C 78 03		JMP OUTPUT	
0366	CA	RDUCE	DEX	NO
0367	CA		DEX	
0368	CA		DEX	
0369	10 E4		BPL TRY	DONE?
036B	4C 78 03		JMP OUTPUT	YES
036E	XX XX XX			UNUSED
0371	XX XX XX			UNUSED
0374	XX XX XX			UNUSED
0377	XX			UNUSED

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
0378	A9 00	OUTPUT	LDA #00	OUTPUT ROUTINE
037A	85 ED		STA ED	INITIALIZE DATPNTL
037C	A9 1D		LDA #1D	$1D_{16} = 29_{10} =$
037E	85 E7		STA TEMPA	3 x NO. OF BINS - 1
0380	85 E8		STA TEMPB	
0382	A9 03	LINE	LDA #03	
0384	85 E9		STA COUNTER	3 CHARACTER COUNTER
0386	A9 2E		LDA #2E	
0388	20 EE 02		JSR CH	
038B	A6 E7	ASCII	LDX TEMPA	
038D	BD 8E 02		LDA TABLE, X	ASCII TABLE ADDRESS = 028E
0390	20 EE 02		JSR CH	PRINT NEXT ASCII CHARACTER
0393	C6 E7		DEC TEMPA	
0395	C6 E9		DEC COUNTER	
0397	D0 F2		BNE ASCII	PROCEED AFTER 3 CHARACTERS
0399	20 33 03		JSR SP	
039C	20 33 03		JSR SP	PRINT 2 SPACES
039F	A9 2E		LDA #2E	ASCII "."
03A1	20 EE 02		JSR CH	PRINT "."
03A4	20 C4 03		JSR OUTBYT	PRINT HI ORDER BYTE
03A7	A5 2D		LDA MHI	STORE HI ORDER BYTE IN NHI
03A9	85 2F		STA NHI	
03AB	20 C4 03		JSR OUTBYT	PRINT LO ORDER BYTE
03AE	20 33 03		JSR SP	PRINT SPACE
03B1	20 C4 03		JSR OUTBYT	PRINT EXPONENT
03B4	20 D0 03		JSR PLOT	PLOT HI ORDER BYTE
03B7	20 BA 00		JSR CR	CARRIAGE RETURN, LINE FEED
03BA	A5 E8		LDA TEMPB	
03BC	10 C4		BPL LINE	PROCEED AFTER 10_{10} LINES
03BE	20 BA 00		JSR CR	CARRIAGE RETURN, LINE FEED
03C1	4C 00 00		JMP MAIN	RETURN TO MAIN WAITING LOOP
03C4	A4 E8	OUTBYT	LDY TEMPB	SUBROUTINE TO

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
03C6	B1 ED	•	LDA (ED), Y	PRINT ONE BYTE
03C8	85 2D		STA MHI	
03CA	20 FA 03		JSR BT	
03CD	C6 E8		DEC TEMPB	
03CF	60		RTS	SUBROUTINE RETURN

ADDRESS	CODE	LABEL	ASSEMBLY	COMMENT
03D0	F8	PLOT	SED	SUBROUTINE TO PLOT
03D1	38		SEC	HI ORDER BYTE (NHI)
03D2	A5 28		LDA EXP	COMPARE EXPONENT
03D4	E5 2D		SBC MHI	WITH MAX. VALUE
03D6	F0 0D		BEQ ORDNAT	IF NON-ZERO STORE
03D8	85 2D		STA MHI	DIFFERENCE IN MHI
03DA	A0 04	SETY	LDY #04	SHIFT HI ORDER BYTE
03DC	46 2F	RTSHFT	LSR NHI	4 BITS RIGHT (MHI)
03DE	88		DEY	TIMES
03DF	D0 FB		BNE RTSHFT	
03E1	C6 2D		DEC MHI	
03E3	D0 F5		BNE SETY	
03E5	D8	ORDNAT	CLD	CLEAR DECIMAL MODE
03E6	A9 2A		LDA #2A	ASCII "*"
03E8	20 EE 02		JSR CH	PRINT "*"
03EB	F8		SED	SET DECIMAL MODE
03EC	38		SEC	
03ED	A5 2F		LDA NHI	DECREMENT HI ORDER
03EF	E9 01		SBC #01	BYTE BY ONE
03F1	90 05		BCC EXIT	RESULT NEGATIVE?
03F3	85 2F		STA NHI	NO, PRINT AGAIN
03F5	4C E5 03		JMP ORDNAT	
03F8	D8	EXIT	CLD	YES, CLEAR DECIMAL MODE
03F9	60		RTS	SUBROUTINE RETURN
03FA	78	BT	SEI	SUBROUTINE TO
03FB	20 3B 1E		JSR PRTBYT	PRINT BYTE
03FE	58		CLI	
03FF	60		RTS	SUBROUTINE RETURN

TECHNICAL REPORT DATA
(Please read Instructions on the reverse before completing)

1. REPORT NO. EPA-600/2-79-140	2.	3. RECIPIENT'S ACCESSION NO.
4. TITLE AND SUBTITLE IMPLEMENTATION OF A MICROCOMPUTER-MODIFIED ELECTRICAL AEROSOL ANALYZER		5. REPORT DATE August 1979 6. PERFORMING ORGANIZATION CODE
7. AUTHOR(S) C. Lewis		8. PERFORMING ORGANIZATION REPORT NO.
9. PERFORMING ORGANIZATION NAME AND ADDRESS Environmental Sciences Research Laboratory - RTP, NC Office of Research and Development U.S. Environmental Protection Agency Research Triangle Park, NC 27711		10. PROGRAM ELEMENT NO. 1AD712 BB-19 (FY-78) 11. CONTRACT/GRANT NO.
12. SPONSORING AGENCY NAME AND ADDRESS Environmental Sciences Research Laboratory - RTP, NC Office of Research and Development U.S. Environmental Protection Agency Research Triangle Park, NC 27711		13. TYPE OF REPORT AND PERIOD COVERED In-house 14. SPONSORING AGENCY CODE EPA/600/09
15. SUPPLEMENTARY NOTES		

16. ABSTRACT

A commercial electrical aerosol analyzer has been modified through addition of a digital voltmeter, a microcomputer and a printer. In unattended operation, the system can acquire a differential particle size distribution, print and plot the resulting spectrum in any of three possible representations, and repeat the cycle at preselected intervals.

Details of the system assembly and instructions for its operation are given. The appendices contain a flow chart and documented listing of Program EAA, the microcomputer program which operates the system.

17. KEY WORDS AND DOCUMENT ANALYSIS

a. DESCRIPTORS	b. IDENTIFIERS/OPEN ENDED TERMS	c. COSATI Field/Group
Air pollution *Aerosols *Analyzers *Particle size distribution *Revisions		13B 07D 14B
18. DISTRIBUTION STATEMENT RELEASE TO PUBLIC	19. SECURITY CLASS *(This Report)* UNCLASSIFIED	21. NO. OF PAGES 50
	20. SECURITY CLASS *(This page)* UNCLASSIFIED	22. PRICE

EPA Form 2220-1 (9-73)